
//          Пример реализации схемы "крест" в двумерном случае 
 
pi=3.141592653; 
L=2;                       // x in [0, L] 
H=1;                      // y in [0, H] 
N=101;                  // число узлов по x 
M=51;                   // число узлов по y 
T=1;                      // длина расчетного промежутка по времени  
hx=L/(N-1);          // шаг по x 
hy=2*H/(2*M-3); // шаг по y 
J=2+round(2*T*sqrt(1/hx^2+1/hy^2)); // число слоев по времени с учетом  
                                                               // условия Куранта 
tau=T/(J-1);       // шаг по t 
x=zeros(N,1);    // массив узлов по x 
y=zeros(M,1);   // массив узлов по y 
t=zeros(J,1);      // массив узлов по t 
// Задание сетки--------------------------------------------------------------- 
for n=1:N 
    x(n)=(n-1)*hx; 
end 
for m=1:M 
    y(m)=(m-1)*hy; 
end 
for j=1:J 
    t(j)=(j-1)*tau; 
end 
//------------------------------------------------------------------------------------ 
// Аналитическое решение---------------------------------------------------- 
u=zeros(M,N,J); 
for j=1:J 
    for n=1:N 
        for m=1:M 
            u(m,n,j)=t(j)*x(n)*y(m)+... 
                cos(pi*sqrt(2)*t(j))*sin(pi*x(n)/2)*sin(pi*y(m)/2); 
        end 
    end 
end 
//----------------------------------------------------------------------------------- 
// Численное решение--------------------------------------------------------- 
v=zeros(M,N,J); 
err=zeros(M,N,J);    // погрешность численного решения 
// Начальные условия: 
for n=1:N 
    for m=1:M 
        v(m,n,1)=sin(pi*x(n)/2)*sin(pi*y(m)/2); 
        v(m,n,2)=x(n)*y(m)*tau+(1-tau^2*pi^2)*v(m,n,1); 
        err(m,n,2)=v(m,n,2)-u(m,n,2); 
    end 
end 
 
for j=2:J-1 
    for n=2:N-1 
        for m=2:M-1 
            v(m,n,j+1)=2*v(m,n,j)-v(m,n,j-1)+... 



                tau^2*4*(v(m,n+1,j)-2*v(m,n,j)+v(m,n-1,j))/hx^2+... 
                tau^2*4*(v(m+1,n,j)-2*v(m,n,j)+v(m-1,n,j))/hy^2; 
            err(m,n,j+1)=v(m,n,j+1)-u(m,n,j+1); 
        end 
    end 
    // граничные условия: 
    for m=1:M 
        v(m,N,j+1)=2*t(j+1)*y(m); 
        err(m,N,j+1)=v(m,N,j+1)-u(m,N,j+1); 
    end 
    for n=2:N-1 
        v(M,n,j+1)=v(M-1,n,j+1)+t(j+1)*x(n)*hy; 
        err(M,n,j+1)=v(M,n,j+1)-u(M,n,j+1); 
    end 
end 
//------------------------------------------------------------------------------ 
 
j=50; 
 
figure 
    surf(x,y,u(:,:,j)) 
    xlabel('x') 
    ylabel('y') 
    zlabel('u') 
    title('analitical solution') 
 
figure 
    surf(x,y,v(:,:,j)) 
    xlabel('x') 
    ylabel('y') 
    zlabel('u') 
    title('calculated solution') 
 
figure 
    surf(x,y,err(:,:,j)) 
    xlabel('x') 
    ylabel('y') 
    zlabel('err') 
    title('error') 
 

 

 

 

 

 

 

 



//          Пример реализации эволюционно-факторизованной схемы для уравнения  
//          колебаний в двумерном случае 
 
pi=3.141592653; 
sigma=1/4;            // параметр в неявной схеме 
L=2;                      // x in [0, L] 
H=1;                      // y in [0, H] 
N=101;                  // число узлов по x 
M=51;                   // число узлов по y 
T=1;                      // длина расчетного промежутка по времени  
hx=L/(N-1);          // шаг по x 
hy=2*H/(2*M-3); // шаг по y 
J=2+round(2*T*sqrt(1/hx^2+1/hy^2));   
//J=100;                                     
tau=T/(J-1);       // шаг по t 
x=zeros(N,1);    // массив узлов по x 
y=zeros(M,1);   // массив узлов по y 
t=zeros(J,1);      // массив узлов по t 
// Задание сетки--------------------------------------------------------------- 
for n=1:N 
    x(n)=(n-1)*hx; 
end 
for m=1:M 
    y(m)=(m-1)*hy; 
end 
for j=1:J 
    t(j)=(j-1)*tau; 
end 
//------------------------------------------------------------------------------------ 
// Аналитическое решение---------------------------------------------------- 
u=zeros(M,N,J); 
for j=1:J 
    for n=1:N 
        for m=1:M 
            u(m,n,j)=t(j)*x(n)*y(m)+... 
                cos(pi*sqrt(2)*t(j))*sin(pi*x(n)/2)*sin(pi*y(m)/2); 
        end 
    end 
end 
//----------------------------------------------------------------------------------- 
// Численное решение--------------------------------------------------------- 
v=zeros(M,N,J);   // численное решение 
err=zeros(M,N,J); // погрешность численного решения 
// Начальные условия: 
for n=1:N 
    for m=1:M 
        v(m,n,1)=sin(pi*x(n)/2)*sin(pi*y(m)/2); 
        v(m,n,2)=x(n)*y(m)*tau+(1-tau^2*pi^2)*v(m,n,1); 
        err(m,n,2)=v(m,n,2)-u(m,n,2); 
    end 
end 
 
 
 



w=zeros(M,N);   //вспомогательная функция при переходе со слоя на слой 
alpha_x=zeros(N-1,1); 
beta_x=zeros(N-1,1); 
alpha_y=zeros(M-1,1); 
beta_y=zeros(M-1,1); 
Ax=4*sigma*tau^2/hx^2; 
Cx=1+2*Ax; 
Ay=4*sigma*tau^2/hy^2; 
Cy=1+2*Ay; 
for j=2:J-1 
    // вычисление вспомогательной функции w --------------------------------  
    for m=2:M-1 
        // прямой ход прогонки по x-------------------------------------------------- 
        for n=2:N-1  
            F=4*(v(m,n-1,j)-2*v(m,n,j)+v(m,n+1,j))/hx^2+... 
                4*(v(m-1,n,j)-2*v(m,n,j)+v(m+1,n,j))/hy^2; 
            alpha_x(n)=Ax/(Cx-Ax*alpha_x(n-1)); 
            beta_x(n)=(F+Ax*beta_x(n-1))/(Cx-Ax*alpha_x(n-1)); 
        end 
        // обратный ход прогонки по x---------------------------------------------- 
        for n=N-1:-1:1 
            w(m,n)=alpha_x(n)*w(m,n+1)+beta_x(n); 
        end 
    end 
    //-вычисление функции v на новом слое по времени------------------------ 
    for n=2:N-1 
        // прямой ход прогонки по y-------------------------------------------------- 
        for m=2:M-1 
            F=tau^2*w(m,n)+Ay*(v(m-1,n,j-1)-2*v(m-1,n,j))-... 
                Cy*(v(m,n,j-1)-2*v(m,n,j))+Ay*(v(m+1,n,j-1)-2*v(m+1,n,j)); 
            alpha_y(m)=Ay/(Cy-Ay*alpha_y(m-1)); 
            beta_y(m)=(F+Ay*beta_y(m-1))/(Cy-Ay*alpha_y(m-1)); 
        end 
        // обратный ход прогонки по y----------------------------------------------- 
        v(M,n,j+1)=(beta_y(M-1)+hy*t(j+1)*x(n))/(1-alpha_y(M-1)); 
        err(M,n,j+1)=v(M,n,j+1)-u(M,n,j+1); 
        for m=M-1:-1:1 
            v(m,n,j+1)=alpha_y(m)*v(m+1,n,j+1)+beta_y(m); 
            err(m,n,j+1)=v(m,n,j+1)-u(m,n,j+1); 
        end 
    end 
    // граничные условия: 
    for m=1:M 
        v(m,N,j+1)=2*t(j+1)*y(m); 
        err(m,N,j+1)=v(m,N,j+1)-u(m,N,j+1); 
    end 
end 
//-------------------------------------------------------------------------------------------- 
 
 
j=100; 
figure 
    surf(x,y,u(:,:,j)) 
    xlabel('x') 



    ylabel('y') 
    zlabel('u') 
    title('analitical solution') 
 
figure 
    surf(x,y,v(:,:,j)) 
    xlabel('x') 
    ylabel('y') 
    zlabel('u') 
    title('calculated solution') 
 
figure 
    surf(x,y,err(:,:,j)) 
    xlabel('x') 
    ylabel('y') 
    zlabel('u') 
    title('error') 
 

 

 


